

Guide de Programmation

Scans de Marché (ProScreener)



SOMMAIRE



Présentation de Proscreeener



Chapitre I : Les notions fondamentales

➤ Accéder à Proscreeener.....	2
➤ Utiliser la fenêtre Palmarès.....	6



Chapitre II : La programmation sur ProScreener

➤ Instruction de recherche et de filtre.....	7
➤ Anticipation du volume.....	9
➤ Recherche multi-périodes.....	9
➤ Recherche multi-valeurs.....	10



Chapitre III : Applications pratiques

➤ Exemples simples.....	12
➤ RSI 1heure : survente.....	12
➤ RSI 1heure : surachat.....	12
➤ Croisement de moyennes mobiles haussier.....	12
➤ Croisement de moyennes mobiles baissier.....	13
➤ Exemples plus élaborés.....	13
➤ RSI et retournement haussier.....	13
➤ RSI et retournement baissier.....	14
➤ Avalement haussier.....	14
➤ Avalement baissier.....	15
➤ Triple écran haussier.....	15
➤ Triple écran baissier.....	16



Glossaire



Présentation de ProScreener

ProScreener est un outil de scan très performant qui vous permettra de passer en revue des marchés entiers :

- à la recherche des valeurs respectant une ou plusieurs conditions que vous aurez définies.
- dans une ou plusieurs unités de temps (ex : 1 minute et 1 heure)
- en temps-réel ou à la clôture de chaque barre avec une précision tick par tick

ProScreener intègre le langage de programmation ProBuilder (dont il vous est conseillé de lire au préalable [le manuel](#)) avec des extensions s'appliquant exclusivement aux conditions de filtrage. Les recherches en temps-réel ou à clôture de la barre, peuvent s'appliquer aux unités de temps suivantes :

- 1 minute
- 2 minutes
- 3 minutes
- 5 minutes
- 10 minutes
- 15 minutes
- 30 minutes
- 1 heure
- 2 heures
- 3 heures
- 4 heures
- Journalier
- Hebdomadaire

Les résultats d'une recherche ProScreener sont affichés par une liste des 50 meilleures valeurs (choisies parmi les résultats) sur la base du critère de tri défini.


Ce document s'inscrit dans le prolongement du [manuel ProBuilder](#) mais peut être lu indépendamment. Grâce à une approche théorique guidée et des exemples concrets, vous acquerrez au fil de votre lecture les bases de la programmation d'indicateurs qui vous permettront, à partir des commandes spécifiques à ProScreener, de définir des filtres sur les différents marchés disponibles sur la plateforme. Vous trouverez, à la fin de ce dernier, un Glossaire qui vous donnera une vue de l'ensemble des commandes utilisables dans ce module.

En vous souhaitant nos meilleurs vœux de réussite, bonne lecture !

Chapitre I : Les notions fondamentales

Accéder à ProScreener

Vous pouvez accéder au module ProScreener en cliquant sur "Affichage" puis "ProScreener", comme montré ci-dessous :



Apparaîtra alors la fenêtre ProScreener, où il est possible de sélectionner un code déjà créé ou bien de procéder à la création d'un nouveau code. Pour ce faire, cliquez sur "Gestion ProScreeners" afin d'accéder à la fenêtre de programmation, qui propose le choix entre :

- la "Programmation assistée", qui permet de définir les conditions du critère de recherche sans programmer (pour en savoir plus sur cette fonctionnalité, nous vous conseillons la vision de la vidéo, "[Créer un ProScreener sans écrire une ligne de code](#)").
- la "Création par programmation", qui permet de créer des codes plus complexes et de définir les paramètres d'application.

La fenêtre de Création par programmation se compose de 4 sections :

1. Zone de programmation
2. Marché d'application
3. Période considérée
4. Tri des résultats



Dans la **première** section vous pouvez :

- Programmer directement un **ProScreener** dans la zone de texte
- Utiliser la fonction d'aide "Insérer Fonction", qui permet de retrouver dans une nouvelle fenêtre la bibliothèque des fonctions disponibles, séparées en sept catégories, afin de vous aider contextuellement lors de la programmation.

Prenons comme exemple l'instruction "**SCREENER**", qui permet d'effectuer la recherche des conditions (disponible dans la section "Commandes ProScreener").

Sélectionnez donc le mot "**SCREENER**" et cliquez sur "Ajouter" : la commande s'ajoutera à la zone de programmation.



La commande "**SCREENER**" vous permet de définir quelle(s) condition(s) rechercher.

La dernière ligne de chaque ProScreener doit commencer par la commande "**SCREENER**".

Supposons que l'on souhaite rechercher toutes les valeurs du SRD France dont l'ouverture de la barre (période) en cours soit supérieure à la clôture de la barre précédente . Nous écrivons donc :

```
c1 = (Open > Close[1])
SCREENER[c1]
```

Une fois le code défini, nous allons choisir dans la **deuxième** section, une liste prédéfinie ou personnalisée sur laquelle effectuer la recherche.

Remarque : La recherche ne peut être effectuée que sur les valeurs d'un même marché. Pour appliquer un ProScreener à une liste personnelle, il faudra veiller à ce que toutes les valeurs incluses dans celle-ci fassent partie du même marché.

La **troisième** section permet de définir la période utilisée dans la recherche du ProScreener. En effet, chaque cours pouvant être visualisé selon une période différente, une figure repérée sous une certaine vue pourrait ne pas exister dans une autre.

La **quatrième** section concerne le tri des résultats selon un critère défini dans le code (Voir section : "Instruction de recherche et de filtre"). Avec des conditions peu restrictives, les résultats d'un ProScreener peuvent être nombreux (plusieurs milliers par exemple sur le NASDAQ). Dans de tels cas, il convient de filtrer les résultats selon un critère au choix et d'afficher :

- ▀ les 50 valeurs les plus élevées (du critère de tri)
- ▀ les 50 valeurs les plus faibles (du critère de tri)

Par exemple : imaginons que votre programme recherche les volumes supérieurs à 20000 sur les valeurs du SRD en journalier. Le nombre de résultats suite à la recherche devrait logiquement excéder 50 résultats. Paramétrer le critère de tri par exemple sur le % de variation des titres vous permettra d'afficher parmi les résultats les 50 valeurs ayant les variations les plus élevées ou les plus faibles.

Une fois ces 4 paramètres définis, cliquez sur "Exécuter ProScreener" pour lancer la recherche. Si des valeurs entrent dans vos conditions, les résultats sont affichés dans une liste comme ci-dessous

ProScreener - Recherche					
Recherche		Gestion ProScreeners			
	Nom	Dernier	%Var	Volume	Veille
	MERSEN	16,760	+0,96%	28 586	16,600
	S.E.B.	52,160	+0,37%	30 635	51,970
	DANONE	55,770	+0,16%	2 256k	55,680
	ALTRAN TECHN.	5,280	+0,00%	413 450	5,280
	LOREAL	121,550	-0,08%	319 498	121,650
	UBISOFT ENTERTAIN	7,950	-0,13%	86 365	7,960
	CLUB MEDITERRANEE	12,380	-0,16%	35 089	12,400
	VEOLIA ENVIRON.	9,977	-0,23%	3 345k	10,000
	METROPOLE TV	11,785	-0,38%	68 481	11,830
	RALLYE	29,500	-0,51%	46 610	29,650
	PERNOD RICARD	93,080	-0,57%	396 181	93,610
	NEOPOST	47,415	-0,61%	54 338	47,705
	SODEXO	70,980	-0,63%	96 238	71,430
	BOUYGUES	20,970	-0,76%	744 160	21,130
	ATOS	50,440	-0,81%	144 827	50,850
	EURAZEO	39,080	-0,84%	32 344	39,410
	PPR	164,85	-0,90%	241 414	166,35
	AIR LIQUIDE	91,460	-0,93%	485 097	92,320
	HAVAS	4,610	-1,01%	133 093	4,657
	CAP GEMINI	33,715	-1,13%	381 985	34,100
	JC DECAUX SA.	19,990	-1,16%	28 088	20,225
	CARREFOUR	20,900	-1,21%	1 227k	21,155
	ACCOR	25,005	-1,22%	785 603	25,315
	AIR FRANCE -KLM	6,915	-1,27%	1 852k	7,004
	SAINT GOBAIN	28,620	-1,31%	1 334k	29,000
	THALES	31,145	-1,39%	98 186	31,585
	KLEPIERRE	30,285	-1,48%	71 536	30,740

✓ Nb résultats : 125 (max 50 affichés)

Utiliser la fenêtre Palmarès

En plus du ProScreener, ProRealTime met à votre disposition deux palmarès. Les palmarès sont des fenêtres de scan en temps réel, dont les critères sont prédéfinis (contrairement au ProScreener, qui est totalement personnalisable). Les deux fenêtres de palmarès sont identiques et vous permettent des recherches simultanées (exemple : variation à la hausse dans une fenêtre, variations à la baisse dans l'autre fenêtre).

Les critères de recherche proposés dans le Palmarès appartiennent à 4 catégories et permettent de détecter :

- Les variations et gaps
- Les plus amples variations de pré-ouverture
- Les principales figures de chandeliers
- Les valeurs de Spread et de volume du carnet d'ordre

Palmarès - Variations - SRD - Variation Hausse

Variations Pré-ouverture Chandeliers Carnet d'ordres

Marché SRD Critère Variation Hausse Depuis Hier

Nb résultats : 27

	Nom	%Var	Veille	Volume	Dernier	Réf prix
EADS		+4,92%		8 597k	39,025	37,2
SINCLAIR IS PHARMA		+3,33%		13 735	0,31	0,3
TOTAL GABON		+1,19%		156	451,10	445,81
TRANSGENE		+1,15%		12 233	8,76	8,66
INTLE PLANT.HEVEAS		+1,08%		2 710	60,65	60
VRANKEN-POMMERY		+1,00%		517	20,20	20
MERSEN		+0,96%		28 769	16,760	16,6
GEMALTO		+0,75%		331 546	65,840	65,35
GROUPE STERIA		+0,73%		101 545	11,010	10,93
BENETEAU		+0,63%		11 294	7,950	7,9
BONDUELLE		+0,61%		8 292	18,220	18,11
FONCIERE DES MURS		+0,57%		1 199	17,65	17,55
GROUPE CRIT		+0,55%		524	14,69	14,61
VEOLIA ENVIRON.		+0,50%		3 450k	10,050	10
THERMADOR GROUPE		+0,48%		2 340	58,94	58,66
SCOR SE		+0,44%		279 718	22,750	22,65
BIOMERIEUX		+0,43%		9 548	72,900	72,59
S.E.B.		+0,37%		34 986	52,160	51,97
CEGEDIM		+0,36%		2 776	24,76	24,67
DANONE		+0,25%		2 316k	55,820	55,68
UBISOFT ENTERTAIN		+0,25%		86 676	7,980	7,96
FRANCE TELECOM		+0,25%		4 552k	7,624	7,6
DNXCORP		+0,25%		1 456	20,45	20,4
EUTELSAT COMMUNIC.		+0,23%		202 663	26,145	26,08
SCHLUMBERGER		+0,13%		5 180	55,23	55,16

Afin d'en savoir plus sur l'utilisation du Palmarès, nous vous conseillons la vision de la vidéo ["Comment utiliser les palmarès"](#)



Chapitre II : La programmation sur ProScreener

Dans ce chapitre vous seront illustrées les 5 commandes ProBuilder propres au ProScreener. Vous pouvez retrouver ces commandes également dans la Bibliothèque des fonctions, à la section "Commandes ProScreener". Nous passerons en revue :

- Instruction de recherche et de filtre
- Anticipation du volume
- Recherche multi-période
- Recherche multi-valeur



Instruction de recherche et de filtre

La commande "**SCREENER**" permet de lancer la recherche. Elle est l'équivalent de "**RETURN**" pour ProBuilder et, comme cette dernière, peut être suivie d'une panoplie de fonctions que nous allons étudier ensemble.

"**SCREENER**" s'utilise de la façon suivante :

SCREENER[Condition]

Prenons un exemple :

```
c1 = (Close < BollingerDown[10] (Close))
SCREENER[c1]
```

Cherchons toutes les valeurs d'un marché dont le prix de clôture est strictement inférieur à la bande de Bollinger inférieure. Les bandes de Bollinger sont calculées sur 10 périodes et appliquées au cours de clôture.

Il est possible d'étendre la recherche à plusieurs conditions, qui peuvent être satisfaites au même temps ou alternativement. Pour ce faire, la syntaxe de ProScreener sera:

SCREENER[Condition1 AND Condition2]

ou bien

SCREENER[Condition1 OR Condition2]

Exemple :

Recherchons les valeurs dont le prix est supérieur au canal de Bollinger et qui présentent une tendance à la hausse.

```
REM Clôture au dessus de la Boll sup
Condition1 = (Close > BollingerUp[20] (Close))
REM Clôture > à l'Ouverture
Condition2 = Close > Open
REM MM7 > MM23
Condition3 = (Average[7] (Close) > Average[23] (Close))
SCREENER[Condition1 AND Condition2 AND Condition3]
```

Il est également possible de définir une constante ou un indicateur (prédéfini ou personnalisé) qui servira à la fois comme critère de filtre et de tri des résultats. En effet :

Si plus de 50 résultats correspondent à la recherche, l'indicateur filtre les valeurs à afficher. Dans ce cas, la quatrième section de la fenêtre de programmation vous permet d'en définir l'ordre de parution (voir [description page 5](#)).

Si moins de 50 résultats correspondent à la recherche, l'indicateur trie les résultats sur la base de ce nouvel indicateur

La syntaxe pour utiliser une constante est :

SCREENER[c1](TypicalPrice)

Pour un indicateur prédéfini, il sera nécessaire de stocker d'abord l'indicateur dans une variable (ici on l'appellera "Criteria") :

Criteria = RSI[14](Close)

SCREENER[c1](Criteria)

Si on fait appel à un indicateur personnalisé, il faudra utiliser l'instruction "CALL", définie dans le [manuel ProBuilder](#).


MyRSI = CALL "RSI" [14]

SCREENER[c1](MyRSI)

Exemple :

Repérons toutes les valeurs d'un marché dont le volume est supérieur à 50000. Créons un indicateur qui nous permet d'évaluer si le RSI est en surachat ou en survente. Nous afficherons les résultats du ProScreener triés sur la base de ce dernier indicateur.

```
c1 = Volume > 50000
IF RSI[20](Close) > 70 THEN
    Criteria = 1
ELSIF RSI[20](Close) < 30 THEN
    Criteria = -1
ENDIF
SCREENER[c1](Criteria AS "sur-transaction")
```



The screenshot shows the 'ProScreener - Filtre Actions' window. It contains a table with the following columns: Nom, %Var, Volume, and sur-transaction. The table lists 16 stocks, with their volume and 'sur-transaction' values. The 'sur-transaction' column is sorted in descending order, with values ranging from 1 to 1. The table also shows the percentage change (%Var) for each stock. At the bottom, it indicates 'Nb résultats : 106 (max 50 affichés)'.

Nom	%Var	Volume	sur-transaction
CHRISTIAN DIOR	-1,94%	71 787	1
ALTRAN TECHN.	-0,57%	536 795	1
EFFAGE	-1,31%	93 009	1
CREDIT AGRICOLE	-2,61%	7 084k	1
ATOS	-1,36%	192 674	1
TELEPERFORMANCE	-2,07%	98 862	1
METROPOLE TV	-0,51%	96 231	1
UBISOFT ENTERTAIN	+0,00%	98 481	1
TF1	-2,74%	204 349	1
RALLYE	-1,42%	60 596	1
ALTEN	-2,28%	74 189	1
IPSOS	-4,56%	57 885	1
SOCIETE GENERALE	-3,42%	3 832k	1
PUBLICIS GROUPE SA	-2,46%	573 345	1
BNP PARIBAS ACT.A	-4,10%	4 350k	1

Il est possible de ne pas utiliser les parenthèses, en introduisant la commande "SORT BY" :

SCREENER[c1] SORT BY TypicalPrice

Si nous souhaitons attribuer un nom personnalisé à cette colonne, nous devons utiliser la fonction "AS", comme illustré ci-dessous :

SCREENER[c1](TypicalPrice AS "typical price")

Anticipation du volume

La commande "**EstimatedVolume**" permet d'avoir une estimation linéaire du volume de la barre courante. Plus précisément, on calcule le volume estimé à partir de la formule suivante :

$$\text{Volume_estimé} = \text{Volume} * \text{Coefficient_Multiplicateur}$$

ou

$$\text{Coefficient_Multiplicateur} = \text{durée de la vue} / \text{temps écoulé depuis que le chandelier se dessine}$$

Cette commande se révèle donc particulièrement intéressante pour comparer le volume estimé au volume réel.

Regardons un exemple :

En vue 10 minutes, si l'on suppose que le dernier chandelier se dessine depuis 1 minute, le volume estimé sera égal à 10 fois le volume actuel.

Cherchons à calculer le rapport du volume anticipé du jour sur le volume de la veille (pour avoir des résultats représentatifs, utiliser une période intraday) :

```
REM Evalue le volume du jour
Vol0 = EstimatedVolume
REM Trouve le volume de la veille
Vol1 = Volume[1]
REM Critère de tri: rapport du volume anticipé du jour sur le volume de la veille
SCREENER (Vol0 / Vol1 AS "Volume")
```

Recherche multi-périodes

Il vous est possible de définir une recherche qui vérifie les critères dans plusieurs unités de temps. Ceci permet de vérifier qu'une condition soit satisfaite par exemple dans le court et le moyen terme. La commande adaptée est "**TIMEFRAME**", et sa syntaxe est la suivante :

TIMEFRAME(unité de temps de la période)

Les différentes unités de temps disponibles se traduisent en code par :

Signification	Code
1 minute	1 minute
2 minutes	2 minutes
3 minutes	3 minutes
5 minutes	5 minutes
10 minutes	10 minutes
15 minutes	15 minutes
30 minutes	30 minutes
1 heure	1 hour
2 heures	2 hours
3 heures	3 hours
4 heures	4 hours
Journalier	daily
Hebdomadaire	weekly

Les instructions qui suivent la ligne **"TIMEFRAME"** seront recherchées seulement dans la période indiquée. Il est donc possible de placer plusieurs lignes **"TIMEFRAME"** dans un même code, afin d'effectuer une recherche multi-périodes.

Prenons un exemple :

On souhaite repérer toutes les valeurs d'Euronext Paris qui vérifient les critères ci-dessous :

- En vue hebdomadaire, l'indicateur Williams %R à 14 périodes a une valeur comprise entre 0 et -20
- En vue 30 minutes, la moyenne mobile exponentielle à 20 périodes vient de croiser à la hausse celle à 12 périodes.

Nous pourrions donc détecter les valeurs en zone de sur-achat en hebdomadaire, mais qui présentent à court terme une tendance baissière.

```
TIMEFRAME(weekly)
Condition1 = Williams[14](Close) < 0 AND Williams[14](Close) > -20
TIMEFRAME(30 minutes)
Condition2 = ExponentialAverage[20](Close) CROSSES OVER ExponentialAverage[12](Close)
SCREENER[Condition1 AND Condition2]
```

Recherche multi-valeurs

Nous avons pu constater qu'avec ProScreener il est nécessaire de spécifier un marché (ou une liste) sur lequel effectuer la recherche.

L'instruction **"EQUITYFRAME"**, permet de mettre en relation le critère avec une valeur spécifique appartenant au même marché de recherche.

Elle s'utilise comme suit :

EQUITYFRAME("nom du marché", "ticker")

La commande permet donc de comparer des résultats à une valeur particulière, mais aussi de construire un nouvel indicateur utilisé comme condition de tri. Pour pouvoir mettre à profit cette commande :

- Doit faire exclusivement appel aux marchés pour lesquels on dispose du temps réel
- Ne peut pas appeler les indices des principaux marchés
- La valeur spécifiée doit faire partie du même marché de recherche (section 2 de la fenêtre ProScreener)

Par exemple, si vous souhaitez afficher 50 valeurs prises au hasard du marché Euronext Paris, avec comme critère le prix de clôture de la barre courante de la valeur AXA (ticker "CS") , vous écrirez :

```
EQUITYFRAME("Actions France", "CS")
MyClose = Close
SCREENER(MyClose)
```


Les différents marchés correspondent à la syntaxe reportée dans le tableau ci-dessous :

Code EQUITYFRAME	Nom Marché	Code EQUITYFRAME	Nom Marché
Actions & ETFs AMEX	Actions "Amex"	Actions Italie	Actions "Italie"
Actions Australie	Actions "Australie"	LSE	London Stock Exchange
Actions Espagne	Actions "Espagne"	Actions NASDAQ	Actions "Nasdaq"
Actions Amsterdam	Euronext Amsterdam	Actions NYSE	Actions "Nyse"
Actions Belgique	Euronext Bruxelles	Actions Suisse VirTX	Actions "Suisse"
Actions Portugal	Euronext Lisbonne	Actions Suisse SWX	Actions "Suisse"
Actions France	Euronext Paris	Actions Allemagne	Actions "Allemagne"
Forex 47 paires	Forex		

Comme pour la commande **"TIMEFRAME"**, tout code qui suivra **"EQUITYFRAME"** sera appliqué à la valeur spécifié par ce dernier. Pour revenir aux données du marché sélectionné dans l'interface ProScreener, nous pouvons écrire :

EQUITYFRAME(default)

Les deux fonctions présentées ci-dessus sont complémentaires et l'exemple suivant l'illustre bien :

Si vous souhaitez afficher sur 50 valeurs prises au hasard du marché Euronext Paris, l'indicateur faisant la différence du prix de clôture de la barre courante de la valeur AXA (ticker "CS") par le prix de clôture respectif des valeurs sélectionnées

```
REM on commence par récupérer les informations à extraire de la valeur AXA
d'Euronext Paris
EQUITYFRAME("Actions France", "CS")
MyClose = Close
REM on revient sur les valeurs du marché sélectionné dans le menu déroulant
"Sélection de la liste"
EQUITYFRAME(default)
REM on récupère les informations que l'on souhaite
CloseVal = MyClose - Close
REM on lance l'affichage
SCREENER(CloseVal AS "MyIndicator")
```

Exemple : Ce screener permet de représenter visuellement la corrélation entre un titre et son marché. On calcule le ratio des deux valeurs sélectionnées. On calcule ensuite la différence de ce ratio par rapport à la veille.

```
TIMEFRAME(daily)
CloseVal = Close
EQUITYFRAME("NASDAQ", "AMZN")
CloseInd = Close
EQUITYFRAME(default)
Ratio = (CloseVal / CloseInd) * 100
ForceRelative = (Ratio - Ratio[1]) * 100
SCREENER(ForceRelative AS "Force Relative")
```


Chapitre III : Applications pratiques

Exemples simples

RSI 1heure : survente

Le RSI est un indicateur de surachat/survente qui peut être précurseur d'un retournement de tendance. Nous allons ici détecter les valeurs qui sont en survente.

La survente est définie par : **RSI < 30**. Elle est d'autant plus marquée que le RSI est proche de zéro. Nous allons donc construire un ProScreener qui renvoie toutes les valeurs dont le RSI est inférieur à 30.

```
REM Calcule le RSI à 14 barres
MyRSI = RSI[14]
REM Filtre : RSI < 30
Filtre = MyRSI < 30
SCREENER[Filtre] (MyRSI AS "RSI")
```

RSI 1heure : surachat

Recherchons les valeurs en surachat (défini par : **RSI > 70**). Il est d'autant plus marqué que le RSI est proche de 100. Nous allons donc construire un screener qui renvoie toutes les valeurs dont le RSI est supérieur à 70.

```
REM Calcule le RSI à 14 barres
MyRSI = RSI[14]
REM Filtre : RSI > 70
Filtre = MyRSI > 70
SCREENER[Filtre] (MyRSI AS "RSI")
```

Croisement de moyennes mobiles haussier

Il s'agit de mettre en évidence une tendance où une moyenne mobile courte croise une moyenne mobile longue.

Nous allons construire un ProScreener qui renvoie toutes les valeurs dont la moyenne mobile à 20 jours croise à la hausse la moyenne mobile à 50 jours.

On calcule par ailleurs le momentum de la différence entre les deux moyennes mobiles afin d'évaluer la puissance du croisement. Si ce nombre est proche de zéro, le croisement est lent et les deux moyennes sont presque parallèles donnant un signal peu significatif. Au contraire, plus ce nombre est élevé et plus le croisement est "dynamique". Un exemple typique est celui d'une moyenne mobile plate croisant une moyenne mobile très nettement orientée à la hausse.

```
REM Calcul de la moyenne mobile à 20 jours
Sma20 = Average[20]
REM Calcul de la moyenne mobile à 50 jours
Sma50 = Average[50]
REM Evalue la vitesse relative de la moyenne mobile courte par rapport à la longue
Vitesse = Momentum(Sma50 - Sma20) * 100 / Close
REM Sélectionne la valeur au moment du croisement
Filtre = Sma20 CROSSES OVER Sma50
SCREENER[Filtre] (Vitesse AS "Dynamique")
```


Croisement de moyennes mobiles baissier

Nous allons construire un ProScreener qui renvoie toutes les valeurs dont la moyenne mobile à 20 jours croise à la baisse la moyenne mobile à 50 jours.

On définit le momentum de la même façon que sur l'exemple précédent.

```
REM Calcul de la moyenne mobile à 20 jours
Sma20 = Average[20]
REM Calcul de la moyenne mobile à 50 jours
Sma50 = Average[50]
REM Evalue la vitesse relative de la moyenne mobile courte par rapport à la
longue
Vitesse = Momentum(Sma20 - Sma50) * 100 / Close
REM Sélectionne la valeur au moment du croisement
Filtre = Sma20 CROSSES UNDER Sma50
SCREENER[Filtre] (Vitesse AS "Dynamique")
```

Exemples plus élaborés

RSI et retournement haussier

Nous vous présentons ici un système permettant de détecter en temps réel les valeurs les plus susceptibles de se retourner et d'amorcer un mouvement de grande envergure.

Traditionnellement les analystes s'intéressaient aux indicateurs de surchauffe ou aux figures de retournement en les isolant de leur contexte (simplement parce qu'ils ne disposaient pas des outils techniques leur permettant de faire mieux).

Aujourd'hui avec le ProScreener il n'y a plus de limitation technique, nous avons donc la possibilité de construire un véritable système de détection basé sur le RSI. Le point de départ est simple: on cherche un RSI en survente qui se retourne à la hausse.

On met donc en place un filtre contenant ces deux conditions ce qui s'écrit en une seule ligne de code :

```
Filtre = RSI < 30 AND Momentum[1] (RSI) > 0
```

Maintenant que nous avons les valeurs susceptibles de nous intéresser, nous allons les classer pour mettre en évidence les placements les plus intéressants. Pour cela, on considère ceux dont la chute a été la plus sévère, comparativement à la volatilité habituelle du titre.

```
REM Filtre les valeurs dont le RSI est en survente et en retournement
// Critère de filtrage : RSI < 30 et croissant
Filtre = RSI < 30 AND Momentum[1] (RSI) > 0
REM Détermine la "force" de la tendance baissière
REM Trouve le point le plus haut sur les 20 barres antérieures
PointHaut = highest[20] (High)
REM Détermine le déclin depuis ce point
Declin = PointHaut - Close
REM Détermine la volatilité habituelle du titre (médiane du True Range sur 3
barres)
Norme = summation[3] (TR) - highest[3] (TR) - lowest[3] (TR)
Vitesse = Declin / Norme
REM Affichage des résultats
SCREENER[Filtre] (Vitesse AS "Puissance baissière")
```


■ RSI et retournement baissier

Le point de départ de ce ProScreener est la recherche d'un RSI en surachat qui se retourne à la baisse. Comme précédemment, mettons en place un filtre. Celui-ci s'écrit en code ProBuilder :

```
Filtre = RSI > 70 AND Momentum[1](RSI) < 0
```

De façon analogue au ProScreener précédent, on récupérera les valeurs dont la hausse a été la plus marquante, comparativement à la volatilité habituelle du titre.

```
REM Filtre les valeurs dont le RSI est en surachat et en retournement
// Critère de filtrage : RSI > 70 et décroissant
Filtre = RSI > 70 AND Momentum[1](RSI) < 0
REM Détermine la "force" de la tendance haussière
// Trouve le point le plus bas sur les 20 barres antérieures
PointBas = lowest[20](Low)
// Détermine la progression des cours depuis ce point
Hausse = Close - PointBas
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
Vitesse = Hausse / Norme
REM Palmarès de screening : les hausses relatives les plus marquantes
SCREENER[Filtre](Vitesse AS "Puissance haussière")
```

■ Avalement haussier

L'avalement haussier est une des figures en chandeliers japonais les plus connues mais elle nécessite évidemment la prise en compte du contexte (existence d'une tendance à retourner).

Commençons par définir ce qu'est un avalement haussier :

- Chandelier précédent de couleur noire
- Chandelier courant ouvrant en-dessous du corps du chandelier précédent
- Chandelier courant clôturant au-dessus du corps du chandelier précédent

Ces trois conditions s'écrivent simplement :

```
Filtre = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
```

Pour ce qui est de la détection de la tendance baissière, nous reprenons le code précédent en resserrant un petit peu l'horizon de temps. En effet, avec le RSI 14, il était raisonnable de s'intéresser à la chute des cours intervenue sur les 20 barres antérieures. Pour ce qui est des chandeliers japonais, l'expérience montre que 8 barres sont suffisantes (ce sont des structures plus réactives et théoriquement limitées à 7 chandeliers).

```
REM Détermine la "force" de la tendance baissière
// Trouve le point le plus haut sur les 8 barres antérieures
PointHaut = highest[8](High)
// Détermine le déclin depuis ce point
Declin = PointHaut - Close
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
REM Critère de filtrage : avalement haussier
Filtre = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
Vitesse = Declin / Norme
REM Palmarès de screening : les déclins relatifs les plus sévères
SCREENER[Filtre](Vitesse AS "Puissance du déclin")
```


■ **Avalement baissier**

Commençons par définir ce qu'est un avalement baissier :

- Chandelier précédent de couleur blanche
- Chandelier courant ouvrant au-dessus du corps du chandelier précédent
- Chandelier courant clôturant en-dessous du corps du chandelier précédent

Ces trois conditions s'écrivent simplement :

Filtre = `Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]`

Pour ce qui est de la détection de la tendance haussière, nous reprenons le code du RSI en retournement baissier.

```
REM détermine la "force" de la tendance haussière
// Trouve le point le plus bas sur les 8 barres antérieures
PointBas = lowest[8](Low)
// Détermine la progression des cours depuis ce point
Hausse = Close - PointBas
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
REM Critère de filtrage : avalement baissier
Filtre = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]
Vitesse = Hausse / Norme
REM Palmarès de screening : les hausses relatives les plus marquantes
SCREENER[Filtre](Vitesse AS "Puissance haussière")
```

■ **Triple écran haussier**

Ce ProScreener est composé de trois conditions sur plusieurs unités de temps :

- **Condition 1** : MACD en vue hebdomadaire < 0
- **Condition 2** : MACD en vue hebdomadaire croissant
- **Condition 3** : Stochastic en vue journalière < 30

On cherche les valeurs plus proches du niveau de l'achat STOP préconisé par le système (dans ce cas, le cours le plus haut de la veille).

Le ProScreener affiche les valeurs qui se trouvent encore sous ce niveau, et celles qui viennent de le franchir et qui restent à proximité (pas plus de +5%).

Les résultats affichés respectent les conditions dans les unités de temps spécifiées (vue hebdomadaire et journalière).

```
REM Condition 1 : MACD weekly < 0 et croissant
TIMEFRAME(weekly)
MyMACD = MACD[12, 26, 9](Close)
c1 = MyMACD < 0 AND MyMACD > MyMACD[1]
REM Condition 2 : Stochastic daily < 30
TIMEFRAME(daily)
MySTO = Stochastic[14, 3](Close)
c2 = MySTO < 30
REM Niveau de l'achat stop
MyStop = High[1]
REM Critère : Position des cours par rapport au STOP
Criteria = (Close / MyStop - 1) * 100
REM Condition 3 : les cours sous le STOP ou à moins de +5%
c3 = Criteria < 5
SCREENER[c1 AND c2 AND c3](Criteria)
```


■ Triple écran baissier

Ce ProScreener est composé de trois conditions sur plusieurs unités de temps :

- **Condition 1** : MACD en vue hebdomadaire > 0
- **Condition 2** : MACD en vue hebdomadaire décroissant
- **Condition 3** : Stochastic en vue journalière > 70

On cherche les valeurs qui sont les plus proches du niveau de la vente STOP préconisée par le système. Ce niveau est le cours le plus bas de la veille.

Le ProScreener affiche les valeurs qui sont encore au-dessus de ce niveau, et celles qui viennent de le franchir et qui restent à proximité (pas plus de -5%), dans le deux unités de temps spécifiées.

```
REM Condition 1 : MACD weekly > 0 et décroissant
TIMEFRAME(weekly)
MyMACD = MACD[12,26,9](Close)
c1 = MyMACD > 0 AND MyMACD < MyMACD[1]
REM Condition 2 : Stochastic daily > 70
TIMEFRAME(daily)
MySTO = Stochastic[14,3](Close)
c2 = MySTO > 70
REM Niveau de la vente stop
MyStop = Low[1]
REM Critère : Position des cours par rapport au STOP
Criteria = (Close / MyStop - 1) * 100
REM Condition 3 : les cours au-dessus du STOP ou à moins de +5%
c3 = Criteria > -5
SCREENER[c1 AND c2 AND c3](Criteria)
```


Glossaire

A

Code	Implementation	Fonction
ABS	ABS(a)	Fonction Mathématique "Valeur Absolue"
AccumDistr	AccumDistr(price)	Désigne l'Accumulation Distribution classique
ADX	ADX[N]	Indicateur Average Directional Index
ADXR	ADXR[N]	Indicateur Average Directional Index Rate
AND	a AND b	Opérateur logique ET
AroonDown	AroonDown[P]	Désigne l'Aroon Down
AroonUp	AroonUp[P]	Désigne l'Aroon Up
ATAN	ATAN(a)	Fonction mathématique "Arc tangente"
AS	RETURN x AS "ResultName"	Instruction servant à nommer une courbe
Average	Average[N](price)	Moyenne Mobile Arithmétique
AverageTrueRange	AverageTrueRange[N](price)	Désigne la moyenne mobile par lissage de Wilder du True Range

B

Code	Implementation	Fonction
BarIndex	BarIndex	Nombre de barres depuis la première barre de données chargée (dans un graphique dans le cas d'un indicateur ProBuilder ou pour un système de trading dans le cas d'un ProBacktest ou ProInvest)
BollingerBandWidth	BollingerBandWidth[N](price)	Bande passante de Bollinger
BollingerDown	BollingerDown[N](price)	Support de la bande de Bollinger
BollingerUp	BollingerUp[N](price)	Résistance de la bande de Bollinger
BREAK	(FOR...DO...BREAK...NEXT) ou (WHILE...DO...BREAK...WEND)	Instruction de sortie forcée de boucle FOR ou WHILE

C

Code	Implementation	Fonction
CALL	myResult = CALL myFunction	Appel de fonction utilisateur
CCI	CCI[N](price) ou CCI[N]	Donne le Commodity Channel Index
ChaikinOsc	ChaikinOsc[Ch1, Ch2](price)	Désigne l'oscillateur de Chaikin
Chandle	Chandle[N](price)	Désigne le Chande Momentum Oscillator
ChandeKrollStopUp	ChandeKrollStopUp[Pp, Qq, X]	Stop de protection selon Chande et Kroll en position acheteuse
ChandeKrollStopDown	ChandeKrollStopDown[Pp, Qq, X]	Stop de protection selon Chande et Kroll en position vendeuse
Close	Close[N]	Désigne le prix de clôture de la barre courante ou de celle n jours auparavant
COLOURED	RETURN x COLOURED(R,G,B)	Colorie une courbe d'une certaine couleur selon la convention RGB
COS	COS(a)	Fonction cosinus
CROSSES OVER	a CROSSES OVER b	Opérateur booléen vérifiant qu'une courbe passe au-dessus d'une autre
CROSSES UNDER	a CROSSES UNDER b	Opérateur booléen vérifiant qu'une courbe passe en dessous d'une autre
cumsum	cumsum(price)	Sommation d'un prix depuis le début de l'historique affiché
CurrentDayOfWeek	CurrentDayOfWeek	Désigne le jour actuel
CurrentHour	CurrentHour	Désigne l'heure actuelle
CurrentMinute	CurrentMinute	Désigne la minute actuelle
CurrentMonth	CurrentMonth	Désigne le mois actuel
CurrentSecond	CurrentSecond	Désigne la seconde actuelle
CurrentTime	CurrentTime	Désigne HeureMinute actuelle
CurrentYear	CurrentYear	Désigne l'année actuelle
CustomClose	CustomClose[N]	Constante paramétrable dans la fenêtre de propriétés
Cycle	Cycle(price)	Indicateur Cycle

D

Code	Implementation	Fonction
Date	Date[N]	Désigne la date de clôture de la barre courante
Day	Day[N]	Jour de clôture de la barre courante
Days	Days[N]	Compteur de jours depuis 1900
DayOfWeek	DayOfWeek[N]	Désigne le jour de la semaine durant lequel la barre courante a clos
DClose	DClose(N)	Prix de clôture de la n-ième journée antérieure à celle de la barre courante
DEMA	DEMA[N](price)	Double Moyenne Mobile Exponentielle
DHigh	DHigh(N)	Prix le plus haut de la n-ième journée antérieure à celle de la barre courante
DI	DI[N](price)	Désigne le Demand Index
DIminus	DIminus[N](price)	Désigne le DI-
DIplus	DIplus[N](price)	Désigne le DI+
DLow	DLow(N)	Prix le plus bas de la n-ième journée antérieure à celle de la barre courante
DO	Voir FOR et WHILE	Instruction facultative des FOR et WHILE pour l'action de bouclage
DOpen	DOpen(N)	Prix d'ouverture de la n-ième journée antérieure à celle de la barre courante
DOWNT0	Voir FOR	Instruction sur boucle FOR pour une lecture décroissante
DPO	DPO[N](price)	Désigne le Detrented Price Oscillator

E

Code	Implementation	Fonction
EaseOfMovement	EaseOfMovement[I]	Désigne l'indicateur Ease of Movement
ELSE	Voir IF/THEN/ELSE/ENDIF	Instruction d'appel de la seconde condition à défaut de la première issue du IF
ELSEIF	Voir IF/THEN/ELSE/ENDIF	Contraction de ELSE IF
EMV	EMV[N]	Désigne l'indicateur Ease of Movement Value
EQUITYFRAME	EQUITYFRAME("marché", "ticker")	Recherche une condition en relation avec une valeur du meme marché
ENDIF	Voir IF/THEN/ELSE/ENDIF	Instruction de clôture des instructions conditionnelles
EndPointAverage	EndPointAverage[N](price)	Moyenne Mobile à dernier point
EstimatedVolume	EstimatedVolume	Indique un volume anticipé
EXP	EXP(a)	Fonction Mathématique "Exponentielle"
ExponentialAverage	ExponentialAverage[N](price)	Moyenne Mobile Exponentielle

F - G

Code	Implementation	Fonction
<code>FOR/TO/NEXT</code>	FOR i =a TO b DO a NEXT	Prend les valeurs désignées du début à la fin ou vice versa
<code>ForceIndex</code>	ForceIndex(price)	Indicateur Force Index déterminant qui contrôle le marché

H

Code	Implementation	Fonction
<code>High</code>	High[N]	Désigne le plus haut cours atteint durant la période N
<code>highest</code>	highest[N](price)	Désigne le plus haut cours sur un horizon donné
<code>HistoricVolatility</code>	HistoricVolatility[N](price)	Désigne la volatilité historique ou statistique
<code>Hour</code>	Hour[N]	Désigne l'heure de clôture de chaque barre

I - J - K

Code	Implementation	Fonction
<code>IF/THEN/ENDIF</code>	IF a THEN b ENDIF	Ensemble d'instructions conditionnelles sans deuxième condition
<code>IF/THEN/ELSE/ENDIF</code>	IF a THEN b ELSE c ENDIF	Ensemble d'instructions conditionnelles
<code>IntradayBarIndex</code>	IntradayBarIndex[N]	Compte le nombre de chandeliers sur le graphique intraday

L

Code	Implementation	Fonction
<code>LinearRegression</code>	LinearRegression[N](price)	Droite de régression linéaire
<code>LinearRegressionSlope</code>	LinearRegressionSlope[N](price)	Pente de la droite de régression linéaire
<code>LOG</code>	LOG(a)	Fonction mathématique "logarithme népérien"
<code>Low</code>	Low[N]	Désigne le plus bas atteint durant la période
<code>lowest</code>	lowest[N](price)	Désigne le plus bas d'une période sur un horizon donné

M

Code	Implementation	Fonction
MACD	MACD[S,L,Sl](price)	Moving Average Convergence Divergence (MACD)
MACDline	MACDLine[S,L](price)	Désigne la ligne du MACD
MassIndex	MassIndex[N]	Indicateur Mass Index appliqué sur N barres
MAX	MAX(a,b)	Fonction mathématique "Maximum"
MedianPrice	MedianPrice	Moyenne du prix le plus haut et du plus bas
MIN	MIN(a,b)	Fonction Mathématique "Minimum"
Minute	Minute	Désigne la minute du moment de la clôture de chaque barre de l'historique
MOD	a MOD b	Fonction Mathématique "Reste de la division euclidienne"
Momentum	Momentum[I]	Momentum (prix de clôture – prix de clôture de la n-ième barre précédente)
MoneyFlow	MoneyFlow[N](price)	Donne le MoneyFlow entre -1 et 1
MoneyFlowIndex	MoneyFlowIndex[N]	Désigne le MoneyFlowIndex
Month	Month[N]	Désigne le mois de la clôture de chaque barre de l'historique

N

Code	Implementation	Fonction
NEXT	Voir FOR/TO/NEXT	Instruction à placer à la fin de la boucle "FOR"
NOT	NOT a	Opérateur logique NON

O

Code	Implementation	Fonction
OBV	OBV(price)	Désigne l' "On-Balance-Volume"
ONCE	ONCE VariableName = VariableValue	Instruction qui en précède une autre qu'on ne veut réaliser qu'une seule fois
Open	Open[N]	Désigne le prix d'ouverture de la barre courante ou celle de n jours auparavant
OR	a OR b	Opérateur logique OU

P - Q

Code	Implementation	Fonction
PriceOscillator	PriceOscillator[S,L](price)	Indicateur Percentage Price oscillator
PositiveVolumeIndex	PriceVolumeIndex(price)	Désigne l'indicateur Positive Volume Index
PVT	PVT(price)	Désigne l'indicateur "Price Volume Trend"

R

Code	Implementation	Fonction
R2	R2[N](price)	Coefficient R Carré (taux d'erreur des prix à la regression linéaire)
Range	Range[N]	Différence entre le prix le plus haut et le plus bas de la barre courante
REM	REM comment	Précède une remarque dans le code
Repulse	Repulse[N](price)	Mesure la poussée haussière et baissière de chaque bougie
RETURN	RETURN Result	Instruction qui renvoie le résultat
ROC	ROC[N](price)	Désigne le "Price Rate of Change"
RSI	RSI[N](price)	Désigne l'oscillateur "Relative Strength Index"
ROUND	ROUND(a)	Fonction mathématique "Arrondi à l'unité"

S

Code	Implementation	Fonction
SAR	SAR[At,St,Lim]	Désigne le Parabolique SAR
SARatdmf	SARatdmf[At,St,Lim](price)	Parreil (parabolique SAR) Désigne le Parabolique SAR ATDMF
SCREENER	SCREENER[c](prix)	Affiche les résultats
SIN	SIN(a)	Fonction Mathématique "Sinus"
SGN	SGN(a)	Fonction Mathématique "Signe de"
SMI	SMI[N,SS,DS](price)	Désigne le Stochastic Momentum Index
SmoothedStochastic	SmoothedStochastic[N,K](price)	Désigne une Stochastique lissée
SORT BY	Screener(c1) SORT BY price	Permet l'affichage ou le tri des résultats
SQUARE	SQUARE(a)	Fonction mathématique "Mise au carré"
SQRT	SQRT(a)	Fonction Mathématique "Mise à la racine carrée"
STD	STD[N](price)	Fonction Statistique "écart-type"
STE	STE[N](price)	Fonction Statistique "écart-erreur"
Stochastic	Stochastic[N,K](price)	Ligne %K de la Stochastique
summation	summation[N](price)	Somme d'un certain prix des N derniers chandeliers
Supertrend	Supertrend[STF,N]	Désigne le Super Trend

T

Code	Implementation	Fonction
TAN	TAN(a)	Fonction mathématique "Tangente"
TEMA	TEMA[N](price)	Moyenne Mobile Exponentielle Triple
THEN	Voir IF/THEN/ELSE/ENDIF	Instruction suivant la première condition de l'instruction "IF"
Time	Time[N]	Donne l'HeureMinuteSeconde et permet de faire appel à l'heure
TimeSeriesAverage	TimeSeriesAverage[N](price)	Moyenne mobile des séries temporelles
TIMEFRAME	TIMEFRAME(P)	Permet de travailler sur une période différente de celle par défaut
TO	Voir FOR/TO/NEXT	Instruction "jusqu'à" dans la boucle "Pour"
Today	Today	Date de la journée actuelle
TotalPrice	TotalPrice[N]	(Clôture + Ouverture + Plus Haut + Plus Bas) / 4
TR	TR(price)	Désigne le True Range
TriangularAverage	TriangularAverage[N](price)	Moyenne Mobile Triangulaire
TRIX	TRIX[N](price)	Triple Moyenne Mobile Exponentielle
TypicalPrice	TypicalPrice[N]	Prix Typique (moyenne de plus haut, plus bas et clôture)

U

Code	Implementation	Fonction
Undefined	a = Undefined	Pour laisser une variable indéfinie (Null)

V

Code	Implementation	Fonction
Variation	Variation(price)	Différence entre la clôture de la veille et la clôture courante en %
Volatility	Volatility[S, L]	Désigne la volatilité de Chaikin
Volume	Volume[N]	Désigne le volume
VolumeOscillator	VolumeOscillator[S,L]	Désigne l'oscillateur de volume
VolumeROC	VolumeROC[N]	Désigne le volume du Rate Of Change

W

Code	Implementation	Fonction
<code>WeightedAverage</code>	<code>WeightedAverage[N](price)</code>	Désigne la Moyenne Mobile Pondérée
<code>WeightedClose</code>	<code>WeightedClose[N]</code>	Moyenne pondérée entre le prix de clôture, le plus haut et la plus bas
<code>WEND</code>	Voir WHILE/DO/WEND	Instruction à placer à la fin de la boucle Tant Que
<code>WHILE/DO/WEND</code>	WHILE (condition) DO (action) WEND	Boucle "Tant Que"
<code>WilderAverage</code>	<code>WilderAverage[N](price)</code>	Donne la moyenne mobile de Wilder
<code>Williams</code>	<code>Williams[N](close)</code>	Calcule le %R de Williams
<code>WilliamsAccumDistr</code>	<code>WilliamsAccumDistr(price)</code>	Indicateur Accumulation/Distribution de Williams

X

Code	Implementation	Fonction
<code>XOR</code>	a XOR b	Opérateur logique OU exclusif

Y

Code	Implementation	Fonction
<code>Year</code>	<code>Year[N]</code>	Donne l'évolution des années
<code>Yesterday</code>	<code>Yesterday[N]</code>	Donne l'évolution du jour d'avant

Z

Code	Implementation	Fonction
<code>ZigZag</code>	<code>ZigZag[Zr](price)</code>	Zig-Zag de la théorie des vagues d'Eliott
<code>ZigZagPoint</code>	<code>ZigZagPoint[Zp](price)</code>	Zig-Zag de la théorie des vagues d'Eliott calculé à Zp points

Autres

Code	Fonction	Code	Fonction
<code>+</code>	Opérateur d'addition	<code><></code>	Opérateur de différence
<code>-</code>	Opérateur de soustraction	<code><</code>	Opérateur d'infériorité strict
<code>*</code>	Opérateur de multiplication	<code>></code>	Opérateur de supériorité strict
<code>/</code>	Opérateur de division décimale	<code><=</code>	Opérateur d'infériorité
<code>=</code>	Opérateur d'égalité	<code>>=</code>	Opérateur de supériorité

10.32	38.78	84.20	75.50	120.57	8.27	21.57	91.27	26.07	70.13	59.75	13.65	35.24	10.32	38.78	84.20	75.50	120.57	8.27	21.57	91.27
-1.06	-0.71	-0.22	+0.28	+1.4	-0.04	-1.71	-0.95	-0.03	+0.54	+1.12	-0.09	-1.46	-1.06	-0.71	-0.22	+0.28	+1.4	-0.04	-1.71	-0.95



ProRealTime.com

— La référence des logiciels de bourse en ligne —

